


/*******************************CRC Analysis.sas*****************************
Description:  coding for infographs data
Date:  Sept 1, 2021
Author:  Nuha Wareg, Jamie Smith, Jane McElroy
Input:  "C:\Users\smithjami\OneDrive - University of Missouri\Desktop\Nuha\sample_dataset_comc_q2_2022.XLSX"  (COMC quarter 2 data) 
Output:

*********************************************************************************/

** Calling in excel data **;
PROC IMPORT OUT = WORK.comc_2022
DATAFILE= "C:\Users\smithjami\OneDrive - University of Missouri\Desktop\Nuha\sample_dataset_comc_q2_2022.XLSX"
DBMS=XLSX REPLACE;
GETNAMES=YES;
RUN;

proc format; 
value fobt  0 = 'no FOBT'
 	  		1 = 'FOBT but not in quarter 2 or 1 year compliant'
	  		2 = 'FOBT = quarter 2 only'
	  		3 = 'FOBT = 1 year compliant only'
	  		4 = 'FOBT = both quarter 2 and 1 year compliant';

value dna 0 = 'no DNA'
 	  		1 = 'DNA but not in quarter 2 or 3 year compliant'
	  		2 = 'DNA = quarter 2 only'
	  		3 = 'DNA = 3 year compliant only'
	  		4 = 'DNA = both quarter 2 and 3 year compliant';

value col   0 = 'no colonoscopy'
 	  		1 = 'colonoscopy but not in quarter 2 or 10-year compliant'
	  		2 = 'colonoscopy = quarter 2 only'
	  		3 = 'colonoscopy = 10 year compliant only'
	  		4 = 'colonoscopy = both quarter 2 and 10 year compliant';

value crc  1 = 'Colonoscopy'
		   2 = 'FIT-FOBT'
		   3 = 'FIT-DNA';

value age 	0 = '< 50 yrs'
            1 = '50-54 yrs'
			2 = '55-59 yrs'
			3 = '60-64 yrs'
			4 = '65-69 yrs'
			5 = '70-75 yrs';
run;

** analysis file **;
data comc2022_q2;
	set comc_2022;
	if exclusion = 'N';
	length provider $35;

** New Variables **;
year = 2022;
quarter = 2;
source = 'COMC';

** collapsing positive/negative responses **;
if fit_dna_result in ('Negative', 'NEG') then fit_dna_result = 'negative';
if fit_fobt_result in ('Negative', 'NEG') then fit_fobt_result = 'negative';
if fit_fobt_result in ('Positive', 'POSITIVE') then fit_fobt_result = 'positive';

** quarterly/yearly compliance indicators **;

** Fit-FOBT quarter 2 and 1 year compliance range **;
if (fit_fobt_date => "01Apr2022"d and fit_fobt_date <= "30Jun2022"d) then fobt_q2_2022 = 1;
if (fit_fobt_date => "01Jul2021"d and fit_fobt_date <= "31MAR2022"d) then fobt_1y_2022 = 1;

** categorizing FOBT compliance **;
if fit_fobt_date = . then fobt_x = 0; else
if fit_fobt_date ne . and fobt_q2_2022 = . and fobt_1y_2022 = . then fobt_x = 1; else
if fit_fobt_date ne . and fobt_q2_2022 = 1 and fobt_1y_2022 = . then fobt_x = 2; else
if fit_fobt_date ne . and fobt_q2_2022 = . and fobt_1y_2022 = 1 then fobt_x = 3; else
if fit_fobt_date ne . and fobt_q2_2022 = 1 and fobt_1y_2022 = 1 then fobt_x = 4;

** Fit-DNA quarter 2 and 3 year compliance range **;
if fit_dna_date ne . then dna = 1;
if (fit_dna_date => "01Apr2022"d and fit_dna_date <= "30Jun2022"d) then dna_q2_2022 = 1;
if (fit_dna_date => "01Jul2019"d and fit_dna_date <= "31MAR2022"d) then dna_3y_2022 = 1;

** categorizing DNA compliance **;
if fit_dna_date = . then dna_x = 0; else
if fit_dna_date ne . and dna_q2_2022 = . and dna_3y_2022 = . then dna_x = 1; else
if fit_dna_date ne . and dna_q2_2022 = 1 and dna_3y_2022 = . then dna_x = 2; else
if fit_dna_date ne . and dna_q2_2022 = . and dna_3y_2022 = 1 then dna_x = 3; else
if fit_dna_date ne . and dna_q2_2022 = 1 and dna_3y_2022 = 1 then dna_x = 4;

** colonoscopy quarter 2 and 10 year compliance range **;
if colonoscopy_date ne . then col = 1;
if (colonoscopy_date => "01Apr2022"d and colonoscopy_date <= "30Jun2022"d) then col_q2_2022 = 1;
if (colonoscopy_date => "01Jul2012"d and colonoscopy_date <= "31MAR2022"d) then col_10y_2022 = 1;

** categorizing colonoscopy compliance **;
if colonoscopy_date = . then col_x = 0; else
if colonoscopy_date ne . and col_q2_2022 = . and col_10y_2022 = . then col_x = 1; else
if colonoscopy_date ne . and col_q2_2022 = 1 and col_10y_2022 = . then col_x = 2; else
if colonoscopy_date ne . and col_q2_2022 = . and col_10y_2022 = 1 then col_x = 3; else
if colonoscopy_date ne . and col_q2_2022 = 1 and col_10y_2022 = 1 then col_x = 4;

** Primary codes (colonoscopy_date = A and Sigmoidoscopy = C and fit_fobt_date = D and fit_dna_date = F) **;
if colonoscopy_date = . and sigmoidoscopy = . and fit_fobt_date = . and fit_dna_date = . then primary_code = 0;
if col_q2_2022 = 1 and (sigmoidoscopy = . and fit_fobt_date = . and fit_dna_date = .) then primary_code = 1;
if fobt_q2_2022 = 1 and (colonoscopy_date = . and sigmoidoscopy = . and fit_dna_date = .) then primary_code = 2;
if dna_q2_2022 = 1  and (colonoscopy_date = . and sigmoidoscopy = . and fit_fobt_date = .) then primary_code = 3;
if (col_10y_2022 = 1 and col_q2_2022 = .) and (sigmoidoscopy = . and fit_fobt_date = . and fit_dna_date = .) then primary_code = 4;
if (fobt_1y_2022 = 1 and fobt_q2_2022 = .) and (colonoscopy_date = . and sigmoidoscopy = . and fit_dna_date = .) then primary_code = 5;
if (dna_3y_2022 = 1 and dna_q2_2022 = .) and (colonoscopy_date = . and sigmoidoscopy = . and fit_fobt_date = .) then primary_code = 6;
if (colonoscopy_date ne . and col_q2_2022 = . and col_10y_2022 = .) and (sigmoidoscopy = . and fit_fobt_date = . and fit_dna_date = .) then primary_code = 7;
if (fit_fobt_date ne . and fobt_q2_2022 = . and fobt_1y_2022 = .) and (colonoscopy_date = . and sigmoidoscopy = . and fit_dna_date = .) then primary_code = 8;
if (fit_dna_date ne . and dna_q2_2022 = . and dna_3y_2022 = .) and (colonoscopy_date = . and sigmoidoscopy = . and fit_fobt_date = .) then primary_code = 9;
if (fobt_q2_2022 = 1 and col_10y_2022 = 1 and col_q2_2022 = .) and (sigmoidoscopy = . and fit_dna_date = .) then primary_code = 21;
if (dna_q2_2022 = 1 and col_10y_2022 = 1 and col_q2_2022 = .) and (sigmoidoscopy = . and fit_fobt_date = .) then primary_code = 22;
if (col_q2_2022 = 1 and fobt_1y_2022 = 1 and fobt_q2_2022 = .) and (sigmoidoscopy = . and fit_dna_date = .) then primary_code = 23;
if (fobt_q2_2022 = 1 and dna_q2_2022 = 1 and col_10y_2022 = 1 and col_q2_2022 = .) and (fit_dna_result in ('positive','negative')) and (source in ('COMC','CHCCMO','Access')) then primary_code = 31;
if (col_q2_2022 = 1 and fobt_q2_2022 = 1 and dna_3y_2022 = 1 and dna_q2_2022 = .) and (fit_dna_result in ('positive','negative')) and (source in ('COMC','CHCCMO','Access')) then primary_code = 32;
if (fobt_q2_2022 = 1 and dna_q2_2022 = 1 and col_10y_2022 = 1 and col_q2_2022 = .) and (fit_dna_result in ('positive','negative')) and (source not in ('COMC','CHCCMO','Access')) then primary_code = 33;
if (col_q2_2022 = 1 and fobt_q2_2022 = 1 and dna_3y_2022 = 1 and dna_q2_2022 = .) and (source not in ('COMC','CHCCMO','Access')) then primary_code = 34;

** adjustments to primary_code above to account for odd cases **;
if primary_code = .  and col_x = 2 then primary_code = 1; ** this is assigning obs where there was a colonoscopy in quarter 2, but other criteria did not fit other code definitions defined above (e.g., FOBT done but not in compliance or within 2 years) **;
if primary_code = .  and col_x = 3 then primary_code = 4; ** this is assigning obs where there was a colonoscopy within 10 years, but other criteria did not fit other code definitions defined above (e.g., FOBT done but not in compliance or within 2 years) **;


** need to create counts here for primary codes 24/25 below **;
compliance_tot = sum(of col_10y_2022 fobt_1y_2022 dna_3y_2022);
quarter_tot = sum(of col_q2_2022 fobt_q2_2022 dna_q2_2022);

if (compliance_tot > 1 and quarter_tot = .) and (source in ('COMC','CHCCMO','Access')) then primary_code = 24;
if (compliance_tot > 1 and quarter_tot = .) and (source not in ('COMC','CHCCMO','Access')) then primary_code = 25;

** calculating some date differences (in weeks) **;
fobt_col_diff  = INTCK('Week',fit_fobt_date, colonoscopy_date);
if fobt_col_diff < 0 then fobt_col_diff = .;

dna_col_diff  = INTCK('Week',fit_dna_date, colonoscopy_date);
if dna_col_diff < 0 then dna_col_diff = .;

**colonoscopy came after fobt or dna **;
if colonoscopy_date > fit_fobt_date then col_after_fobt = 1;
if (fit_fobt_date = . or colonoscopy_date = .) then col_after_fobt = .;

if colonoscopy_date > fit_dna_date then col_after_dna = 1; 
if (fit_dna_date = . or colonoscopy_date = .) then col_after_dna = .;

** Secondary Codes (note: uses some variables derived above) **;
if primary_code in (21,22,23,24,25,31,32,33,34) then secondary_code = 40;
if fit_fobt_result = 'negative' and fobt_col_diff  <= 52 and fobt_col_diff ne . then secondary_code = 41;
if fit_dna_result = 'negative' and dna_col_diff  <= 156 and dna_col_diff ne . then secondary_code = 42;
if (fit_fobt_result = 'positive' or fit_dna_result = 'positive') and (col_after_fobt = 1 or col_after_dna = 1) then secondary_code = 43;


if primary_code in (21,22,23,24,25) and ((fobt_col_diff ne . and fobt_col_diff <= 1) or (dna_col_diff ne . and dna_col_diff <= 1)) then secondary_code = 44;

** assigning current CRC codes **;
if primary_code in (1,4) then current_crc = 1; else
if primary_code in (2,5) then current_crc = 2; else
if primary_code in (3,6) then current_crc = 3; 

** calculating patient age based on last test date in system **;
last_date = max (of colonoscopy_date fit_fobt_date fit_dna_date);

age = round((last_date-date_of_birth)/365.25,.1);

** fixing provider names (note: this may need done each time) **;
if usual_provider = 'ATKINS, KRISTI' then provider = 'Atkins, Kristi';
if usual_provider = 'BAGBY, MARK' then provider = 'Bagby, Mark';
if usual_provider = 'BAILEY, KELVIN' then provider = 'Bailey, Kelvin';
if usual_provider = 'BOWMAN, ELIZABETH' then provider = 'Bowman, Elizabeth';
if usual_provider = 'CLARY, MICHELE' then provider = 'Clary, Michele';
if usual_provider = 'GULLEY, VICKY' then provider = 'Gulley, Vicky';
if usual_provider = 'Galloway, John' then provider = 'Galloway, John';
if usual_provider = 'JONES, JUANITA' then provider = 'Jones, Juanita';
if usual_provider = 'Jones, Jeffrey' then provider = 'Jones, Jeffrey';
if usual_provider = 'KING, L' then provider = 'King, Lisa';
if usual_provider = 'KING, LISA' then provider = 'King, Lisa';
if usual_provider = 'KRANTZ, T' then provider = 'Krantz, Terry';
if usual_provider = 'KRANTZ, TERRY' then provider = 'Krantz, Terry';
if usual_provider = 'Kirkover, Michael' then provider = 'Kirkover, Michael';
if usual_provider = 'MACIAS, MARIO' then provider = 'Macias, Mario';
if usual_provider = 'Nagy, Virginia' then provider = 'Nagy, Virginia';
if usual_provider = 'Neis, Miles' then provider = 'Neis, Miles';
if usual_provider = 'Powell, Jennifer' then provider = 'Powell, Jennifer';
if usual_provider = 'RADEMAN ALONSO, ERIN' then provider = 'Rademan Alonso, Erin';
if usual_provider = 'ROBERTSON, CHANDLER' then provider = 'Robertson, Chandler';
if usual_provider = 'SCHMIDT, DANIEL' then provider = 'Schmidt, Daniel';
if usual_provider = 'SLUYTER, ERIN' then provider = 'Sluyter, Erin';
if usual_provider = 'THOMAS, JAMIE' then provider = 'Thomas, Jamie';
if usual_provider = 'Thomas, Thomas' then provider = 'Thomas, Thomas';
if usual_provider = 'VANDIVER, AMBER' then provider = 'Vandiver, Amber';
if usual_provider = 'VOGL, DANIEL' then provider = 'Vogl, Daniel';
if usual_provider = 'WARREN, BARTON' then provider = 'Warren, Barton';
if usual_provider = 'Wang, Michael' then provider = 'Wang, Michael';
if usual_provider = 'Whitman, Mary' then provider = 'Whitman, Mary';
if usual_provider = 'coon, errica' then provider = 'Coon, Errica';
if usual_provider = 'z-BUCHANAN, SARAH' then provider = 'Buchanan, Sarah';
if usual_provider = 'z-SPARR, LAURIE' then provider = 'Sparr, Laurie';
if usual_provider = 'z-WARREN, SARA' then provider = 'Warren, Sara';
if usual_provider = 'zREYNOLDS, EVAN' then provider = 'Reynolds, Evan';
if usual_provider = 'zzzMATTOX, SHAYNA' then provider = 'Mattox, Shayna';


** age category **;
if age < 50 then agegrp = 0; else
if 50 le age lt 55 then agegrp = 1; else
if 55 le age lt 60 then agegrp = 2; else
if 60 le age lt 65 then agegrp = 3; else
if 65 le age lt 70 then agegrp = 4; else
if age ge 70 then agegrp = 5;

format fobt_x fobt. dna_x dna. col_x col. current_crc crc. last_date mmddyy10. agegrp age.;
run; 


** Creating a count of patients seen by provider **;
proc sort data=comc2022_q2; by provider; run;

data provider_count (keep=provider pat_total);
	length provider $35;
	set comc2022_q2;
	by provider;
	
if first.provider then pcount = 1; else pcount+1; ** creating a count of records **;

if last.provider then pat_total = pcount; ** getting total # of patients **;

if pat_total = .  then delete; ** reducing file to one obs per provider **;
run;

** merging total patient counts back into file **;
proc sort data=provider_count; by provider; run;

data comc2022_final;
	length provider $35;
	merge comc2022_q2(in=keeper) provider_count;
	by provider;
	if keeper;

** creates a variable that denotes providers who have less than 10 patients in file **;
if pat_total < 10 then pt_less10 = 1;

run;


** Earlier - Requested Tables (note: these include all providers and do not restrict by age) **;
proc tabulate data=comc2022_final missing;
class primary_code location;
table all location, n colpctn primary_code*(n rowpctn colpctn);
where primary_code in (0,1,2,3);
run;

proc tabulate data=comc2022_final missing;
class current_crc crc_status;
table all current_crc, n colpctn crc_status*(n rowpctn colpctn);
run;

proc freq data=comc2022_final;
table fit_dna_result*primary_code/list missing;
where primary_code in (3,6,9,22,33); ** note: there were two additional DNA tests (both neg), but the primary code was a 24 **;
run;

proc freq data=comc2022_final;
tables age;
where secondary_code = 40;
run;

proc tabulate data=comc2022_final missing out=provider_output;  ** this code outputs the data needed for the provider data below **;
class provider crc_status;
tables all provider, n crc_status*(n rowpctn);
where pt_less10 ne 1;
run;


** Creating datasets for provider output/database **;
** total **;
data CRC_T(keep=provider crc_n);
	set provider_output;
	where CRC_STATUS = '';

** renaming variables **;
crc_n = N;

if provider = '' then provider = 'Total';

label
crc_n = 'Total # of Patients';

run;

** CRC = No **;
data CRC_N(keep=provider crc_n_n crc_n_pct);
	set provider_output;
	where CRC_STATUS = 'N';

** renaming variables **;
crc_n_n = N;

crc_n_pct = pctn_10;
if pctn_10 = . then crc_n_pct = pctn_00;
crc_n_pct = round((crc_n_pct),.1);

if provider = '' then provider = 'Total';

label
crc_n_n 	 = 'CRC Status No (N)'
crc_n_pct    = 'CRC Status No (%)';

run;


** CRC = Yes **;
data CRC_Y(keep=provider crc_y_n crc_y_pct crc_y_plus10 id);
	set provider_output;
	where CRC_STATUS = 'Y';

** renaming variables **;
crc_y_n = N;

crc_y_pct = pctn_10;
if pctn_10 = . then crc_y_pct = pctn_00;
crc_y_pct = round((crc_y_pct),.1);

crc_y_plus10 = round((crc_y_n*.10),.1); ** # patients needed if increased by 10% **;

if provider = '' then provider = 'Total';

label
crc_y_n 	 = 'CRC Status Yes (N)'
crc_y_pct    = 'CRC Status Yes (%)'
crc_y_plus10 = '# of patients needed for 10% increase'
provider     = 'Provider Name';

if provider = 'Total' then id = 1; else id = 2;
run;

** sorting files prior to merge **;
proc sort data = crc_t; by provider; run;
proc sort data = crc_n; by provider; run;
proc sort data = crc_y; by provider; run;

data crc_providers;
	merge crc_t crc_n crc_y;
	by provider;
run;

proc sort data=crc_providers; by id; run;

** final file for providers **;
data crc_provider_info (drop=id);
	set crc_providers;
run;

proc sort data=crc_provider_info out=plot_sort; by descending crc_y_pct; run;  ** sorting so that highest percentage first **;

** exporting new file **;
proc export data=crc_provider_info
	outfile='C:\Users\smithjami\OneDrive - University of Missouri\Desktop\Nuha\crc_provider_2022_q2.xlsx'
	dbms=xlsx replace;
run;;


** Scatter plot for Q2 2022 CRC Provider Screening Rates **;
ods rtf file = 'C:\Users\smithjami\OneDrive - University of Missouri\Desktop\CRC preliminary results for infographics(09-01-22).doc'; ** location of word file, adjust as needed **;
proc print data=crc_provider_info label noobs;
run;

ods graphics / reset=all border=off width=600px height=400px;
proc sgplot data=plot_sort noborder;
   scatter y=crc_y_pct x=provider/markerattrs=(color=black size=10 symbol=squarefilled);
   yaxis display=(nolabel) min=0 max=80;
   xaxis label = 'Q2 2022 CRC Screening Rates of Providers' display=(novalues noticks);
   where provider ne 'Total';
   title1 justify=RIGHT "National Goal - 80%" ;
   
run;

** Infosheet - 2022 (Q2) CRC Screening Data for Patients Aged 50-75 **;
proc tabulate data=comc2022_final missing; 
class crc_status current_crc agegrp race sex_at_birth location;
tables all crc_status current_crc agegrp race sex_at_birth, n colpctn location*(n colpctn);
tables all current_crc agegrp race sex_at_birth location, n colpctn crc_status*(n colpctn); 
where agegrp ne 0;
title '2022 (Q2) CRC Screening Data for Patients Aged 50-75';
run;

proc logistic descending data=comc2022_final ;
class sex_at_birth(ref='F') agegrp(ref='50-54 yrs') location(ref='Camdenton');  ** assign reference group here instead of dummy coding **;
model CRC_status = sex_at_birth agegrp location/risklimits ;
where agegrp ne 0 and location in ('Camdenton','Osage Beach','Richland');
title 'Logistic Model Results - CRC Status 2002(Q2)';
run;

ods rtf close;

